

# **CLASS PLOTENVIRONMENT**

## **User Guide**

Class PLOTENVIRONMENT is a SIMULA class designed to facilitate plotting of curves on a line printer.

Much flexibility is available:

- (1) More than one line printer plot may be produced at a time.
- (2) Each plot may be composed of several curves.
- (3) The user can specify scales for each curve or just let PLOTENVIRONMENT choose the scales so that all data are represented on the plot.

An outline of the class is given below.

```

SIMULATION class PLOTENVIRONMENT;
begin

  LINK class PRINTERPLOT;
  begin
    procedure PLOT(SYMBOL,X,Y); character SYMBOL; real X,Y; ... ;

    procedure PRINT(XSTEP); real XSTEP; ... ;

    real procedure XMIN; ... ;
    real procedure XMAX; ... ;
    real procedure YMIN(SYMBOL); character SYMBOL; ... ;
    real procedure YMAX(SYMBOL); character SYMBOL; ... ;

    procedure SETXMIN(X); real X; ... ;
    procedure SETXMAX(X); real X; ... ;
    procedure SETYMIN(SYMBOL,Y); character SYMBOL; real Y; ... ;
    procedure SETYMAX(SYMBOL,Y); character SYMBOL; real Y; ... ;

    procedure SETTITLE(TITLE); value TITLE; text TITLE; ... ;

    procedure SETKEY(SYMBOL,KEY);
    value KEY; character SYMBOL; text KEY; ... ;

    procedure SETXLABEL(ID); value ID; text ID; ... ;

    procedure SCALING(TYPE); integer TYPE; ... ;

    procedure RENEW; ... ;

    procedure LINESPERPLOT(N); integer N; ... ;
    procedure LINESPERMESH(N); integer N; ... ;

    LINESPERPLOT(101); LINESPERMESH(10);
  end;

  procedure PLOT(SYMBOL,X,Y); character SYMBOL; real X,Y;
  SYSPLOT.PLOT(SYMBOL,X,Y);

  ref(PRINTERPLOT) SYSPLOT;      ref(PRINTFILE) PLOTFILE;
  SYSPLOT:-new PRINTERPLOT;      PLOTFILE:-SYSOUT;

  inner;
  TERMINATE:
    for "all PRINTERPLOT's not PRINTed" do PRINT(0);
end;

```

The class `PRINTERPLOT` is used to produce the line printer plots. Each object of the class represents a single plot.

The most essential attributes of the class are the two procedures `PLOT` and `PRINT`.

Procedure `PLOT` is used for collecting the plot points. To identify the curves on the plot, each plot point must be associated with a non-blank character. By calling `PLOT(SYMBOL,X,Y)`, the plot point  $(X,Y)$  is collected and is associated with the character `SYMBOL`.

Example: (plotting a circle)

```
for V:=0 step 0.01 until 2*PI do PLOT('*',COS(V),SIN(V));
```

Procedure `PRINT` is used for printing the collected information. By calling `PRINT(XSTEP)` a plot is produced with an interval of `XSTEP` units on the X-axis between printed lines. `PRINT(XSTEP)` results in a plot consisting of  $\text{ABS}((X_{\text{MAX}}-X_{\text{MIN}})/X_{\text{STEP}})+1$  lines, where `XMIN` and `XMAX` denote the lower and upper limit for the X-values of the collected points. If `XSTEP` is zero, the number of lines will be as specified by the procedure `LINESPERPLOT` (default is 101 lines).

`PRINT(0)` is automatically executed prior to program termination if the user has not called `PRINT`.

The printing progresses line by line with each line consisting of three components. The first is the X-value associated with that line. The second component of a line contains 101 print positions corresponding to the plot points themselves. The third component is a duplicate field that indicates any instances in which two or more different plot characters occupy the same print position. If this happens, the first character processed is plotted and the duplicate field contains this character followed by the characters that should occupy the same print position. If the duplicate field becomes full, an asterisk is written rightmost in the line.

To facilitate the reading, the plot is provided with a grid in which the distance between grid lines is as specified by the procedure `LINESPERMESH` (default is 10 lines). The grid may be suppressed by calling with a non-positive parameter value.

`YMIN(SYMBOL)` and `YMAX(SYMBOL)` denote the lower and upper limit for the Y-values of the collected points having the plot character `SYMBOL`. When the plot is printed only those points  $(X,Y)$  for which

$$X_{\text{MIN}} \leq X \leq X_{\text{MAX}} \quad \text{and} \quad Y_{\text{MIN}}(\text{SYMBOL}) \leq Y \leq Y_{\text{MAX}}(\text{SYMBOL})$$

are included.

By calling the procedures `SETXMIN`, `SETXMAX`, `SETYMIN` and `SETYMAX` the user can limit the plot during the collection phase as well as the printing phase. Limits that are not fixed by the user will automatically be updated during the collection.

The plot may be given a title by calling the procedure `SETTITLE`.

Procedure SETKEY associates a plot symbol with explanatory text (KEY, stripped to 9 characters). The key description precedes the plot.

The procedure SETXLABEL can be used to associate the X-axis with explanatory text (ID, stripped to 11 characters).

The procedure SCALING may be used to specify which type of scaling is desired for the Y-values when the plot is printed. The actual value of the integer parameter TYPE has the following meaning:

TYPE = 1 : Each curve is scaled individually and "readable" scale values are printed on the Y-axis. A readable linear scale is defined here as a scale with interval size a product of an integer power of 10 and 1, 2 or 5, and scale values integer multiples of the interval size. This type of scaling is the default.

TYPE >= 2: All curves are given the same scale and "readable" scale values are printed on the Y-axes. Those curves scaled individually by the user (using SETYMIN or SETYMAX) are not touched.

TYPE = -1: As TYPE=1 except that no attempt is made to achieve "readable" scale values.

TYPE <=-2: As TYPE>=1 except that no attempt is made to achieve "readable" scale values.

(TYPE=0 is interpreted as TYPE=1).

RENEW is a procedure which can be used to erase all points collected in a plot. User-fixed limits, if any, are preserved.

Usually the plot is printed on SYSOUT. Note however, that the user may use an alternative PRINTFILE by letting PLOTFILE reference an open PRINTFILE.

A pre-defined PRINTERPLOT-object, called SYSPLOT, makes it very easy to produce a plot. The user collects points by calling the global procedure PLOT. Printing is automatically carried out before the program terminates (with a default length of 101 lines).

## ERROR MESSAGES

In case of illegal use of the facilities described, an error message is output and the program is terminated. The possible error messages are listed below.

```
PLOT(SYMBOL, , ):  SYMBOL = ' '
SETYMIN(SYMBOL, ):  SYMBOL = ' '
SETYMAX(SYMBOL, ):  SYMBOL = ' '
SETKEY(SYMBOL, ):  SYMBOL = ' '
YMIN(SYMBOL):  SYMBOL = ' '
YMAX(SYMBOL):  SYMBOL = ' '
LINESPERPLOT(N):  N<=1
PRINT( ):  PLOTFILE.IMAGE.LENGTH < 101+2*WIDTH
```

where WIDTH is the field width for numbers in the plot (default value is 11, corresponding to 5 significant digits).